content: -- |
  -- Module      : Benchmarks.VectorStreams
  -- Copyright   : (c) 2018 Harendra Kumar
  --               (c) 2018 Philipp Schuster
  --
  -- License     : MIT
  -- Maintainer  : harendra.kumar@gmail.com
  
  {-# LANGUAGE ScopedTypeVariables #-}
  {-# LANGUAGE TemplateHaskell #-}
  
  module Benchmarks.VectorStreams where
  
  import Benchmarks.DefaultMain (defaultMain)
  import Benchmarks.Common (value, maxValue, appendValue)
  import Prelude
         (Monad, Int, (+), ($), (.), return, even, (>), (<=),
          subtract, undefined, Maybe(..))
  import qualified Prelude as P
  
  import qualified Data.Vector.Fusion.Stream.Monadic as S
  
  -------------------------------------------------------------------------------
  -- Stream generation and elimination
  -------------------------------------------------------------------------------
  
  type Stream m a = S.Stream m a
  
  {-# INLINE source #-}
  source :: Monad m => Int -> Stream m Int
  --source n = S.fromList [n..n+value]
  source n = S.unfoldrM step n
      where
      step cnt =
          if cnt > n + value
          then return Nothing
          else return (Just (cnt, cnt + 1))
          {-
  source n = S.unfoldr step n
      where
      step cnt =
          if cnt > n + value
          then Nothing
          else (Just (cnt, cnt + 1))
              -}
  
  {-# INLINE sourceN #-}
  sourceN :: Monad m => Int -> Int -> Stream m Int
  sourceN count begin = S.unfoldrM step begin
      where
      step i =
          if i > begin + count
          then return Nothing
          else return (Just (i, i + 1))
  
  {-# INLINE sourceIntFromThenTo #-}
  sourceIntFromThenTo :: Monad m => Int -> Stream m Int
  sourceIntFromThenTo n = S.enumFromStepN n 1 value
  
  -------------------------------------------------------------------------------
  -- Append
  -------------------------------------------------------------------------------
  
  {-# INLINE appendSourceR #-}
  appendSourceR :: Int -> P.IO ()
  appendSourceR n =
      toNull $ P.foldr (S.++) S.empty (P.map S.singleton [n..n+appendValue])
  
  {-# INLINE appendSourceL #-}
  appendSourceL :: Int -> P.IO ()
  appendSourceL n =
      toNull $ P.foldl (S.++) S.empty (P.map S.singleton [n..n+appendValue])
  
  -------------------------------------------------------------------------------
  -- Elimination
  -------------------------------------------------------------------------------
  
  {-# INLINE runStream #-}
  runStream :: Monad m => Stream m a -> m ()
  runStream = S.mapM_ (\_ -> return ())
  
  {-# INLINE toNull #-}
  {-# INLINE toList #-}
  {-# INLINE foldl #-}
  {-# INLINE last #-}
  toNull :: Monad m => Stream m Int -> m ()
  toList :: Monad m => Stream m Int -> m [Int]
  foldl  :: Monad m => Stream m Int -> m Int
  last   :: Monad m => Stream m Int -> m Int
  
  toNull = runStream
  toList = S.toList
  foldl  = S.foldl' (+) 0
  last   = S.last
  
  -------------------------------------------------------------------------------
  -- Transformation
  -------------------------------------------------------------------------------
  
  {-# INLINE transform #-}
  transform :: Monad m => Stream m a -> m ()
  transform = runStream
  
  {-# INLINE composeN #-}
  composeN
      :: Monad m
      => Int -> (Stream m Int -> Stream m Int) -> Stream m Int -> m ()
  composeN n f =
      case n of
          1 -> transform . f
          2 -> transform . f . f
          3 -> transform . f . f . f
          4 -> transform . f . f . f . f
          _ -> undefined
  
  {-# INLINE scan #-}
  {-# INLINE map #-}
  {-# INLINE mapM #-}
  {-# INLINE filterEven #-}
  {-# INLINE filterAllOut #-}
  {-# INLINE filterAllIn #-}
  {-# INLINE takeOne #-}
  {-# INLINE takeAll #-}
  {-# INLINE takeWhileTrue #-}
  {-# INLINE dropOne #-}
  {-# INLINE dropAll #-}
  {-# INLINE dropWhileTrue #-}
  {-# INLINE dropWhileFalse #-}
  scan, map, mapM,
      filterEven, filterAllOut, filterAllIn,
      takeOne, takeAll, takeWhileTrue,
      dropOne, dropAll, dropWhileTrue, dropWhileFalse
      :: Monad m => Int -> Stream m Int -> m ()
  
  scan           n = composeN n $ S.scanl' (+) 0
  map            n = composeN n $ S.map (+1)
  mapM           n = composeN n $ S.mapM return
  filterEven     n = composeN n $ S.filter even
  filterAllOut   n = composeN n $ S.filter (> maxValue)
  filterAllIn    n = composeN n $ S.filter (<= maxValue)
  takeOne        n = composeN n $ S.take 1
  takeAll        n = composeN n $ S.take maxValue
  takeWhileTrue  n = composeN n $ S.takeWhile (<= maxValue)
  dropOne        n = composeN n $ S.drop 1
  dropAll        n = composeN n $ S.drop maxValue
  dropWhileFalse n = composeN n $ S.dropWhile (> maxValue)
  dropWhileTrue  n = composeN n $ S.dropWhile (<= maxValue)
  
  -------------------------------------------------------------------------------
  -- Iteration
  -------------------------------------------------------------------------------
  
  iterStreamLen, maxIters :: Int
  iterStreamLen = 10
  maxIters = 100000
  
  {-# INLINE iterateSource #-}
  iterateSource
      :: Monad m
      => (Stream m Int -> Stream m Int) -> Int -> Int -> Stream m Int
  iterateSource g i n = f i (sourceN iterStreamLen n)
      where
          f (0 :: Int) m = g m
          f x m = g (f (x P.- 1) m)
  
  {-# INLINE iterateMapM #-}
  {-# INLINE iterateScan #-}
  {-# INLINE iterateFilterEven #-}
  {-# INLINE iterateTakeAll #-}
  {-# INLINE iterateDropOne #-}
  {-# INLINE iterateDropWhileFalse #-}
  {-# INLINE iterateDropWhileTrue #-}
  iterateMapM, iterateScan, iterateFilterEven, iterateTakeAll, iterateDropOne,
      iterateDropWhileFalse, iterateDropWhileTrue :: Monad m => Int -> Stream m Int
  
  -- Scan increases the size of the stream by 1, drop 1 to not blow up the size
  -- due to many iterations.
  iterateScan n = iterateSource (S.drop 1 . S.scanl' (+) 0) maxIters n
  iterateMapM n = iterateSource (S.mapM return) maxIters n
  iterateFilterEven n = iterateSource (S.filter even) maxIters n
  iterateTakeAll n = iterateSource (S.take maxValue) maxIters n
  iterateDropOne n = iterateSource (S.drop 1) maxIters n
  iterateDropWhileFalse n = iterateSource (S.dropWhile (> maxValue)) maxIters n
  iterateDropWhileTrue n = iterateSource (S.dropWhile (<= maxValue)) maxIters n
  
  -------------------------------------------------------------------------------
  -- Mixed Composition
  -------------------------------------------------------------------------------
  
  {-# INLINE scanMap #-}
  {-# INLINE dropMap #-}
  {-# INLINE dropScan #-}
  {-# INLINE takeDrop #-}
  {-# INLINE takeScan #-}
  {-# INLINE takeMap #-}
  {-# INLINE filterDrop #-}
  {-# INLINE filterTake #-}
  {-# INLINE filterScan #-}
  {-# INLINE filterMap #-}
  scanMap, dropMap, dropScan, takeDrop, takeScan, takeMap, filterDrop,
      filterTake, filterScan, filterMap
      :: Monad m => Int -> Stream m Int -> m ()
  
  scanMap    n = composeN n $ S.map (subtract 1) . S.scanl' (+) 0
  dropMap    n = composeN n $ S.map (subtract 1) . S.drop 1
  dropScan   n = composeN n $ S.scanl' (+) 0 . S.drop 1
  takeDrop   n = composeN n $ S.drop 1 . S.take maxValue
  takeScan   n = composeN n $ S.scanl' (+) 0 . S.take maxValue
  takeMap    n = composeN n $ S.map (subtract 1) . S.take maxValue
  filterDrop n = composeN n $ S.drop 1 . S.filter (<= maxValue)
  filterTake n = composeN n $ S.take maxValue . S.filter (<= maxValue)
  filterScan n = composeN n $ S.scanl' (+) 0 . S.filter (<= P.maxBound)
  filterMap  n = composeN n $ S.map (subtract 1) . S.filter (<= maxValue)
  
  -------------------------------------------------------------------------------
  -- Zipping and concat
  -------------------------------------------------------------------------------
  
  {-# INLINE zip #-}
  zip :: Monad m => Stream m Int -> m ()
  zip src = transform $ (S.zipWith (,) src src)
  
  {-# INLINE concatMap #-}
  concatMap :: Monad m => Stream m Int -> m ()
  concatMap src = transform $ (S.concatMap (S.replicate 3) src)
  
  main :: P.IO ()
  main = $(defaultMain "VectorStreams")
  